USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83–875

Book 6

MODELING TECHNIQUES

Module Documentation for the Evapotranspiration Package

The Evapotranspiration Package (EVT1) consists of four modules, all of which are called by the MAIN program. The modules are:

EVT1AL — Allocates space for arrays to contain maximum ET rate (EVTR), surface elevation (SURF), extinction depth (EXDP), and, if option 2 is specified, the layer indicator (IEVT).

EVT1RP — Reads arrays containing the maximum ET rate (in terms of a volume per unit area), surface elevation, extinction depth, and, if option 2 is specified, the layer indicator. Maximum ET rates are multiplied by cell area to get the maximum ET for each node as a volumetric rate.

EVT1FM — Determines, for each horizontal location, which cell is at the surface. Determines if there is ET from that cell. If there is ET, add the appropriate terms to HCOF and RHS.

EVT1BD — Calculates the rates and accumulated volume of ET out of the flow system.

Narrative for Module EVT1AL

This module allocates space in the X array to store data relating to evapotranspiration.

1. Print a message identifying the package.

2. Read and print the option indicator (NEVTOP) and the unit number for cell-by-cell flow terms (IEVTCB).

3. See if the ET option (NEVTOP) is legal. If NEVTOP is illegal (not 1 or 2), print a message saying the option is illegal. Do not allocate storage. STOP.

4. If NEVTOP is legal, print NEVTOP.

5. If the cell-by-cell flow terms are to be recorded, print the unit number (IEVTCB) where they will be recorded.

6. Allocate space for the maximum ET-rate array (EVTR), the extinction-depth array (EXDP), and the ET-surface array (SURF).

7. If the ET option (NEVTOP) is equal to two, allocate space for a layer-indicator array (IEVT).

8. Calculate and print the number of elements in the X array used by the ET package.

9. RETURN.

Flow Chart for Module EVT1AL

NEVTOP is the ET option.

If NEVTOP = 1, ET is from the top layer.

If NEVTOP = 2, ET is from the layer specified by the user in the indicator array (IEVT).

IEVTCB is the unit number on which cell-by-cell flow terms for ET will be written.

EVTR is an array which contains the maximum ET rate for each horizontal cell location.

SURF is an array which contains the elevation of the ET surface.

EXDP is an array which contains the extinction depth for ET.

IEVT is an array which contains the layer number from which ET is taken for each horizontal location. It is used only if option 2 has been specified.



```
      SUBROUTINE EVT1AL(ISUM,LENX,LCIEVT,LCEVTR,LCEXDP,LCSURF,
     1                  NCOL,NROW,NEVTOP,IN,IOUT,IEVTCB)
C
C-----VERSION 1607 12MAY1987 EVT1AL
C     ******************************************************************
C     ALLOCATE ARRAY STORAGE FOR EVAPOTRANSPIRATION
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
C     ------------------------------------------------------------------
C
C1------IDENTIFY PACKAGE.
      WRITE(IOUT,1)IN
    1 FORMAT(1HO,'EVT1 -- EVAPOTRANSPIRATION PACKAGE, VERSION 1,',
     1     ' 9/1/87',' INPUT READ FROM UNIT',I3)
C
C2------READ NEVTOP AND IEVTCB.
      READ(IN,3)NEVTOP,IEVTCB
    3 FORMAT(2I10)
C
C3------CHECK TO SEE THAT ET OPTION IS LEGAL.
      IF(NEVTOP.GE.1.AND.NEVTOP.LE.2)GO TO 200
C
C3A-----IF ILLEGAL PRINT A MESSAGE & ABORT SIMULATION.
      WRITE(IOUT,8)
    8 FORMAT(1X,'ILLEGAL ET OPTION CODE. SIMULATION ABORTING')
      STOP
C
C4------IF THE OPTION IS LEGAL THEN PRINT THE OPTION CODE.
  200 IF(NEVTOP.EQ.1) WRITE(IOUT,201)
  201 FORMAT(1X,'OPTION 1 -- EVAPOTRANSPIRATION FROM TOP LAYER')
      IF(NEVTOP.EQ.2) WRITE(IOUT,202)
  202 FORMAT(1X,'OPTION 2 -- EVAPOTRANSPIRATION FROM ONE SPECIFIED',
     1      ' NODE IN EACH VERTICAL COLUMN')
      IRK=ISUM
C
C5------IF CELL-BY-CELL TERMS TO BE SAVED THEN PRINT UNIT NUMBER.
      IF(IEVTCB.GT.0) WRITE(IOUT,203) IEVTCB
  203 FORMAT(1X,'CELL-BY-CELL FLOW TERMS WILL BE SAVED ON UNIT',I3)
C
C6------ALLOCATE SPACE FOR THE ARRAYS EVTR, EXDP AND SURF.
      LCEVTR=ISUM
      ISUM=ISUM+NCOL*NROW
      LCEXDP=ISUM
      ISUM=ISUM+NCOL*NROW
      LCSURF=ISUM
      ISUM=ISUM+NCOL*NROW
C
C7------IF OPTION 2 THEN ALLOCATE SPACE FOR THE INDICATOR ARRAY(IEVT)
      LCIEVT=ISUM
      IF(NEVTOP.NE.2)GO TO 300
      ISUM=ISUM+NCOL*NROW
C
C8------CALCULATE & PRINT AMOUNT OF SPACE USED BY ET PACKAGE.
  300 IRK=ISUM-IRK
      WRITE(IOUT,4)IRK
    4 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED FOR EVAPOTRANSPIRATION')
      ISUM1=ISUM-1
      WRITE(IOUT,5)ISUM1,LENX
    5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
      IF(ISUM1.GT.LENX)WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE MADE LARGER***')
C
C9------RETURN.
      RETURN
      END
```

List of Variables for Module EVT1AL

Variable	Range	Definition
IEVTCB	Package	Flag. If IEVTCB > 0 and ICBCFL ≠ 0, cell-by-cell flow terms for the EVT1 Package will be recorded on UNIT = IEVTCB.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IRK	Module	Before this module allocates space, IRK is set equal to ISUM. After allocation, IRK is subtracted from ISUM to get the amount of space in the X array allocated by this module.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	Index number of the last element of the X array allocated by this module.
LCEVTR	Package	Location in the X array of the first element of array EVTR.
LCEXDP	Package	Location in the X array of the first element of array EXDP.
LCIEVT	Package	Location in the X array of the first element of array IEVT.
LCSURF	Package	Location in the X array of the first element of array SURF.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
NCOL	Global	Number of columns in the grid.
NEVTOP	Package	ET option: = 1, ET is from the top layer. = 2, ET at each horizontal-cell location is from the layer specified by the user in the layer-indicator array (IEVT).
NROW	Global	Number of rows in the grid.

Narrative for Module EVT1RP

This module reads data used to calculate the terms which represent evapotranspiration.

1. Read the values INSURF, INEXDP, INEVTR, and INIEVT which indicate whether the data contained in arrays SURF, EXDP, EVTR, and IEVT, respectively, used during the last stress period, are to be used for the current stress period.

2. Test INSURF to see where the ET-surface array (SURF) is coming from. If INSURF is less than zero, the ET-surface elevation used in the last stress period will be used again in this stress period. Print a message to that effect and GO TO 4.

3. INSURF is greater than or equal to zero. CALL U2DREL to read SURF.

4. Test INEVTR to see where the maximum ET rate (EVTR) is coming from. If INEVTR is less than zero, the maximum ET rate used in the last stress period will be used again in this stress period. Print a message to that effect and GO TO 7.

5. INEVTR is greater than or equal to zero. CALL U2DREL to read the maximum ET rate (EVTR).

6. Multiply the maximum ET rate by the area to get a volumetric rate.

7. Test INEXDP to see where the extinction rate is coming from. If INEXDP is less than zero, the extinction depth used in the last stress period will be used again in this stress period. Print a message to that effect and GO TO 9.

8. If INEXDP is greater than or equal to zero, CALL U2DREL to read the extinction depth.

9. If the ET option (NEVTOP) is equal to two, a layer-indicator array is needed.

10. Test INIEVT to see where the layer indicator is coming from. If INIEVT is less than zero, the indicator array used in the last stress period will be used again in this stress period. Print a message to that effect and GO TO 12.

11. If INIEVT is greater than or equal to zero, CALL U2DINT to read the IEVT array.

12. RETURN.

Flow Chart for Module EVT1RP

INEVTR is a flag which, when set, indicates that the maximum ET rate EVTR should be read for the current stress period. If it is clear (less than zero), maximum ET rates from the last stress period should be reused.

INIEVT, INSURF, and INEXDP are flags similar to INEVTR used for the layer indicator array (IEVT), the ET surface array (SURF), and the extinction depth array (EXDP), respectively.

EVTR is an array containing the maximum ET rate for every horizontal cell location.

SURF is an array containing the ET surface elevation for each horizontal cell location.

EXDP is an array containing the extinction depth for each horizontal cell location.

IEVT is an array containing a layer indicator for each horizontal cell location. For each horizontal cell location, it indicates the layer number of the cell at that location from which ET is taken. It is used only if the ET option (NEVTOP) is equal to two.

NEVTOP is the ET option.

If NEVTOP = 1, ET is from the top layer.

If NEVTOP = 2, ET is from the layer specified by the user in the indicator array (IEVT).



```
      SUBROUTINE EVT1RP(NEVTOP,IEVT,EVTR,EXDP,SURF,DELR,DELC,
     1                  NCOL,NROW,IN,IOUT)
C
C-----VERSION 1635 24JUL1987 EVT1RP
C     ******************************************************************
C     READ EVAPOTRANSPIRATION DATA
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 ANAME
      DIMENSION IEVT(NCOL,NROW),EVTR(NCOL,NROW),EXDP(NCOL,NROW),
     1          SURF(NCOL,NROW),ANAME(6,4),DELR(NCOL),DELC(NROW)
C
      DATA ANAME(1,1),ANAME(2,1),ANAME(3,1),ANAME(4,1),ANAME(5,1),
     1  ANAME(6,1) /'    ',' ',' ',' ET',' LAY','ER I','NDEX'/
      DATA ANAME(1,2),ANAME(2,2),ANAME(3,2),ANAME(4,2),ANAME(5,2),
     1  ANAME(6,2) /'    ',' ',' ',' ET',' SUR','FACE'/
      DATA ANAME(1,3),ANAME(2,3),ANAME(3,3),ANAME(4,3),ANAME(5,3),
     1  ANAME(6,3) /' EVA','POTR','ANSP','IRAT','ION ','RATE'/
      DATA ANAME(1,4),ANAME(2,4),ANAME(3,4),ANAME(4,4),ANAME(5,4),
     1  ANAME(6,4) /'    ',' ','EXTI','NCTI','ON D','EPTH'/
C     ------------------------------------------------------------------
C
C1------READ FLAGS SHOWING WHETHER DATA IS TO BE REUSED.
      READ(IN,6)INSURF,INEVTR,INEXDP,INIEVT
    6 FORMAT(4I10)
C
C2------TEST INSURF TO SEE WHERE SURFACE ELEVATION COMES FROM.
      IF(INSURF.GE.0)GO TO 32
C
C2A-----IF INSURF<0 THEN REUSE SURFACE ARRAY FROM LAST STRESS PERIOD
      WRITE(IOUT,3)
    3 FORMAT(1HO,'REUSING SURF FROM LAST STRESS PERIOD')
      GO TO 35
C
C3------IF INSURF=>0 THEN CALL MODULE U2DREL TO READ SURFACE.
   32 CALL U2DREL(SURF,ANAME(1,2),NROW,NCOL,0,IN,IOUT)
C
C4------TEST INEVTR TO SEE WHERE MAX ET RATE COMES FROM.
   35 IF(INEVTR.GE.0)GO TO 37
C
C4A-----IF INEVTR<0 THEN REUSE MAX ET RATE.
      WRITE(IOUT,4)
    4 FORMAT(1HO,'REUSING EVTR FROM LAST STRESS PERIOD')
      GO TO 45
C
C5------IF INEVTR=>0 CALL MODULE U2DREL TO READ MAX ET RATE.
   37 CALL U2DREL(EVTR,ANAME(1,3),NROW,NCOL,0,IN,IOUT)
C
C6------MULTIPLY MAX ET RATE BY CELL AREA TO GET VOLUMETRIC RATE
      DO 40 IR=1,NROW
      DO 40 IC=1,NCOL
      EVTR(IC,IR)=EVTR(IC,IR)*DELR(IC)*DELC(IR)
   40 CONTINUE
C
C7------TEST INEXDP TO SEE WHERE EXTINCTION DEPTH COMES FROM
   45 IF(INEXDP.GE.0)GO TO 47
C
C7A-----IF INEXDP<0 REUSE EXTINCTION DEPTH FROM LAST STRESS PERIOD
      WRITE(IOUT,5)
    5 FORMAT(1HO,'REUSING EXDP FROM LAST STRESS PERIOD')
      GO TO 48
C
C8------IF INEXDP=>0 CALL MODULE U2DREL TO READ EXTINCTION DEPTH
   47 CALL U2DREL(EXDP,ANAME(1,4),NROW,NCOL,0,IN,IOUT)
C
C9------IF OPTION(NEVTOP) IS 2 THEN WE NEED AN INDICATOR ARRAY.
  48  IF(NEVTOP.NE.2)GO TO 50
C
C10-----IF INIEVT<0 THEN REUSE LAYER INDICATOR ARRAY.
      IF(INIEVT.GE.0)GO TO 49
      WRITE(IOUT,2)
    2 FORMAT(1HO,'REUSING IEVT FROM LAST STRESS PERIOD')
      GO TO 50
C
C11-----IF INIEVT=>0 THEN CALL MODULE U2DINT TO READ INDICATOR ARRAY.
   49 CALL U2DINT(IEVT,ANAME(1,1),NROW,NCOL,0,IN,IOUT)
C
C12-----RETURN
   50 RETURN
      END
```

List of Variables for Module EVT1RP

Variable	Range	Definition
ANAME	Module	Label for printout of the input array.
DELC	Global	DIMENSION (NROW), Cell dimension in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimension in the row direction. DELR(J) contains the width of column J.
EVTR	Package	DIMENSION (NCOL,NROW), Maximum ET rate.
EXDP	Package	DIMENSION (NCOL,NROW), Extinction depth.
IC	Module	Index for columns.
IEVT	Package	DIMENSION (NCOL,NROW), Layer number for each horizontal cell location from which ET will be taken if the ET option (NEVTOP) is equal to two.
IN	Package	Primary unit number from which input for this package will be read.
INEVTR	Module	Flag. $\geq$ 0, EVTR array will be read. < 0, EVTR array already in memory from the last stress period will be used.
INEXDP	Module	Flag. $\geq$ 0, EXDP array will be read. < 0, EXDP array already in memory from the last stress period will be used.
INIEVT	Module	Flag. $\geq$ 0, IEVT array will be read. < 0, IEVT array already in memory from the last stress period will be used.
INSURF	Module	Flag. $\geq$ 0, SURF array will be read. < 0, SURF array already in memory from the last stress period will be used.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
NCOL	Global	Number of columns in the grid.
NEVTOP	Package	ET option. = 1, ET is from the top layer. = 2, ET at each horizontal-cell location is from the layer specified in the layer-indicator array (IEVT).
NROW	Global	Number of rows in the grid.
SURF	Package	DIMENSION (NCOL,NROW), Elevation of the ET surface.

Narrative for Module EVT1FM

This module adds terms representing ET to the finite-difference equations.

1. For each horizontal-cell location, determine which layer ET comes from and add the appropriate terms to the equation for the cell. DO STEPS 1-7.

2. Set the layer index equal to one.

3. If option 2 was invoked, get the layer index from the indicator array (IEVT).

4. If the cell is external, move on to the next horizontal-cell location. SKIP STEPS 5-7.

5. If the head in the aquifer is greater than or equal to the ET-surface elevation, add EVTR to RHS and move on to the next horizontal-cell location. SKIP STEPS 6 AND 7.

6. If the head in the aquifer is less than the extinction elevation (ET surface minus extinction depth), no terms need to be added to the finite-difference equation. Move on to the next horizontal-cell location. SKIP STEP 7.

7. Add the term -EVTR/EXDP to HCOF and subtract the term -EVTR(EXDP - SURF)/EXDP from RHS.

8. RETURN.

Flow Chart for Module EVT1FM

IEVT is an array containing a layer indicator for each horizontal cell location. For each horizontal cell location, it indicates the layer number of the cell at that location from which ET is taken. It is used only if the ET option (NEVTOP) is equal to two.

SURF is an array containing the maximum ET rate for every horizontal cell location.

EVTR is an array containing the maximum ET rate for every horizontal cell location.

RHS is an accumulator in which the right hand side of the finite-difference equation is formulated.

HCOF is an accumulator in which a coefficient of head in the finite-difference equation is formulated.

NEVTOP is the ET option.

If NEVTOP = 1, ET is from the top layer.

If NEVTOP = 2, ET is from the layer specified by the user in the indicator array (IEVT).



```
      SUBROUTINE EVT1FM(NEVTOP,IEVT,EVTR,EXDP,SURF,RHS,HCOF,
     1                  IBOUND,HNEW,NCOL,NROW,NLAY)
C
C-----VERSION 1031 10APR1985 EVT1FM
C     ******************************************************************
C        ADD EVAPOTRANSPIRATION TO RHS AND HCOF
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DOUBLE PRECISION HNEW
      DIMENSION IEVT(NCOL,NROW),EVTR(NCOL,NROW),EXDP(NCOL,NROW),
     1          SURF(NCOL,NROW),RHS(NCOL,NROW,NLAY),
     2          HCOF(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY),
     3          HNEW(NCOL,NROW,NLAY)
C     ------------------------------------------------------------------
C
C1------PROCESS EACH HORIZONTAL CELL LOCATION
      DO 10 IR=1,NROW
      DO 10 IC=1,NCOL
C
C2------SET THE LAYER INDEX EQUAL TO 1
      IL=1
C
C3------IF OPTION 2 IS SPECIFIED THEN GET LAYER INDEX FROM IEVT ARRAY
      IF(NEVTOP.EQ.2)IL=IEVT(IC,IR)
C
C4------IF THE CELL IS EXTERNAL IGNORE IT.
      IF(IBOUND(IC,IR,IL).LE.0)GO TO 10
      C=EVTR(IC,IR)
      S=SURF(IC,IR)
      H=HNEW(IC,IR,IL)
C
C5------IF AQUIFER HEAD IS GREATER THAN OR EQUAL TO SURF, ET IS CONSTANT
      IF(H.LT.S) GO TO 5
C
C5A-----SUBTRACT -EVTR FROM RHS
      RHS(IC,IR,IL)=RHS(IC,IR,IL) + C
      GO TO 10
C
C6------IF DEPTH TO WATER>=EXTINCTION DEPTH THEN ET IS 0
    5 D=S-H
      X=EXDP(IC,IR)
      IF(D.GE.X)GO TO 10
C
C7------LINEAR RANGE. ADD ET TERMS TO BOTH RHS AND HCOF.
      RHS(IC,IR,IL)=RHS(IC,IR,IL)+C-C*S/X
      HCOF(IC,IR,IL)=HCOF(IC,IR,IL)-C/X
   10 CONTINUE
C
C8------RETURN
      RETURN
      END
```

List of Variables for Module EVT1FM

Variable	Range	Definition
C	Module	Maximum ET rate.
D	Module	Depth to water.
EVTR	Package	DIMENSION (NCOL,NROW), Maximum ET rate.
EXDP	Package	DIMENSION (NCOL,NROW), Extinction depth.
H	Module	Head in the cell.
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in the cell (J,I,K) in the finite-difference equation.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IEVT	Package	DIMENSION (NCOL,NROW), Layer number, for each horizontal-cell location, from which ET will be taken if the ET option (NEVTOP) is equal to two.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
NCOL	Global	Number of columns in the grid.
NEVTOP	Package	ET option. = 1, ET is from the top layer. = 2, ET at each horizontal cell location is from the layer specified in the layer-indicator array (IEVT).
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of finite-difference equation. RHS is an accumulation of terms from several different packages.
S	Module	ET surface elevation for a cell.
SURF	Package	DIMENSION (NCOL,NROW), Elevation of the ET surface.
X	Module	Extinction depth for a cell.

Narrative for Module EVT1BD

This module calculates rates and volumes removed from the aquifer by evapotranspiration.

1. Clear the rate accumulator RATOUT.

2. If budget terms will be saved, clear the buffer (BUFF) in which they will be accumulated.

3. Process each horizontal-cell location one at a time calculating flow to evapotranspiration (STEPS 4-11).

4. Set the layer index (IL) equal to one.

5. If option 2 is in effect, get the layer index from the layer-indicator array (IEVT).

6. If the cell is external (IBOUND $\leq$ 0), bypass processing of the cell.

7. If the head in the aquifer is greater than the elevation of the ET surface, set the ET rate for the cell equal to the maximum ET rate. SKIP STEPS 8 AND 9.

8. If the depth to the water is greater than the extinction depth, bypass further processing of this cell. SKIP STEP 9.

9. Calculate the ET flow into the model using the linear approximation.

10. Subtract the ET flow from the accumulator (RATOUT).

11. If the cell-by-cell flow terms are to be saved, add the ET rate to the buffer (BUFF).

12. If the cell-by-cell flow terms are to be saved, call module UBUDSV to write the buffer (BUFF) onto a disk.

13. Move RATOUT into the VBVL array for printing by BAS1OT.

14. Add RATOUT multiplied by the time-step length to the volume accumulators in VBVL for printing by BAS1OT.

15. Move the ET budget-term labels to VBNM for printing by BAS1OT.

16. Increment the budget-term counter (MSUM).

17. RETURN.

Flow Chart for Module EVT1BD

RATOUT is an accumulator to which all flows out of the aquifer are added.

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

IEVT is an array containing a layer indicator for each horizontal cell location. For each horizontal cell location, it indicates the layer number of the cell at that location from which ET is taken. It is used only if NEVTOP is equal to two.

SURF is an array containing the ET surface elevation for each horizontal cell location.

Q is the flow to ET from an individual cell.

VBVL is a table of budget entries calculated by component-of-flow packages for use in calculating the volumetric budget.

VBNM is a table of labels for budget terms.



```
      SUBROUTINE EVT1BD(NEVTOP,IEVT,EVTR,EXDP,SURF,IBOUND,HNEW,
     1            NCOL,NROW,NLAY,DELT,VBVL,VBNM,MSUM,KSTP,KPER,
     2            IEVTCB,ICBCFL,BUFF,IOUT)
C-----VERSION 1608 12MAY1987 EVT1BD
C     ******************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR EVAPOTRANSPIRATION
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 VBNM,TEXT
      DOUBLE PRECISION HNEW
      DIMENSION IEVT(NCOL,NROW),EVTR(NCOL,NROW),EXDP(NCOL,NROW),
     1          SURF(NCOL,NROW),IBOUND(NCOL,NROW,NLAY),
     2          VBVL(4,20),VBNM(4,20),HNEW(NCOL,NROW,NLAY),
     3          BUFF(NCOL,NROW,NLAY)
      DIMENSION TEXT(4)
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /'    ','    ','    ','  ET'/
C     ------------------------------------------------------------------
C
C1------CLEAR THE RATE ACCUMULATOR.
      RATOUT=0
C
C2------IF CELL-BY-CELL FLOW TERMS WILL BE SAVED THEN CLEAR THE BUFFER.
      IBD=0
      IF(IEVTCB.LE.0 .OR. ICBCFL.EQ.0) GO TO 5
      IBD=1
      DO 4 IL=1,NLAY
      DO 4 IR=1,NROW
      DO 4 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
    4 CONTINUE
C
C3------PROCESS EACH HORIZONTAL CELL LOCATION
    5 DO 10 IR=1,NROW
      DO 10 IC=1,NCOL
C
C4------SET THE LAYER INDEX EQUAL TO 1
      IL=1
C
C5------IF OPTION 2 IS SPECIFIED THEN GET LAYER INDEX FROM IEVT ARRAY
      IF(NEVTOP.EQ.2)IL=IEVT(IC,IR)
C
C6------IF CELL IS EXTERNAL THEN IGNORE IT.
      IF(IBOUND(IC,IR,IL).LE.0)GO TO 10
      C=EVTR(IC,IR)
      S=SURF(IC,IR)
```

```
      H=HNEW(IC,IR,IL)
C
C7------IF AQUIFER HEAD => SURF,SET Q=MAX ET RATE
      IF(H.LT.S) GO TO 7
      Q=-C
      GO TO 9
C
C8------IF DEPTH=>EXTINCTION DEPTH, ET IS 0
    7 X=EXDP(IC,IR)
      D=S-H
      IF(D.GE.X)GO TO 10
C
C9------LINEAR RANGE . Q=-EVTR(H-EXEL)/EXDP
      Q=C*D/X-C
C
C10-----ACCUMULATE TOTAL FLOW RATE
    9 RATOUT=RATOUT-Q
C
C11-----IF CELL-BY-CELL FLOW TERMS TO BE SAVED THE ADD Q TO BUFFER.
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=Q
   10 CONTINUE
C
C12-----IF C-B-C TO BE SAVED CALL MODULE UBUDSV TO RECORD THEM.
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IEVTCB,BUFF,NCOL,NROW,
     1                         NLAY,IOUT)
C
C13-----MOVE TOTAL ET RATE INTO VBVL FOR PRINTING BY BAS1OT.
      VBVL(3,MSUM)=0.
      VBVL(4,MSUM)=RATOUT
C
C14-----ADD ET(ET_RATE TIMES STEP LENGTH) TO VBVL
      VBVL(1,MSUM)=0.
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
C
C15-----MOVE BUDGET TERM LABELS TO VBNM FOR PRINT BY MODULE BAS1OT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C16-----INCREMENT BUDGET TERM COUNTER
      MSUM=MSUM+1
C
C17-----RETURN
      RETURN
      END
```

List of Variables for Module EVT1BD

Variable	Range	Definition
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
C	Module	Maximum ET rate at a cell.
D	Module	Depth to water below the ET surface.
DELT	Global	Length of the current time step.
EVTR	Package	DIMENSION (NCOL,NROW), Maximum ET rate.
EXDP	Package	DIMENSION (NCOL,NROW), Extinction depth.
H	Module	Head in the cell.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBD	Module	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be recorded for the current time step.
IEVT	Package	DIMENSION (NCOL,NROW), Layer number for each horizontal-cell location from which ET will be taken if the ET option (NEVTOP) is equal to two.
IEVTCB	Package	Flag. If IEVTCB > 0 and ICBCFL ≠ 0, cell-by-cell flow terms for the EVT1 Package will be recorded on UNIT = IEVTCB.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
NCOL	Global	Number of columns in the grid.
NEVTOP	Package	ET option. = 1, ET is from the top layer. = 2, ET at each horizontal-cell location is from the layer specified in the layer-indicator array (IEVT).
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
Q	Module	Flow from ET into the cell. (Reverse the sign to get the flow to ET.)

List of Variables for Module EVT1BD (Continued)

Variable	Range	Definition
RATOUT	Module	Accumulator for the total flow out of the flow field to ET.
S	Module	Elevation of the ET surface for a cell.
SURF	Package	DIMENSION (NCOL,NROW), Elevation of the ET surface.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during simulation. (4,N), Volume out of the flow field during simulation.
X	Module	Extinction depth for a cell.

CHAPTER 11

GENERAL-HEAD BOUNDARY PACKAGE

Conceptualization and Implementation

The function of the General-Head Boundary (GHB) Package is mathematically similar to that of the River, Drain and ET Packages, in that flow into or out of a cell i,j,k, from an external source is provided in proportion to the difference between the head in the cell, $h_{i,j,k}$, and the head assigned to the external source, $h_{bi,j,k}$. Thus a linear relationship between flow into the cell and head in the cell is established, i.e.

$$Q_{bi,j,k} = C_{bi,j,k} (h_{bi,j,k} - h_{i,j,k}) \qquad (78)$$

where $Q_{bi,j,k}$ is the flow into cell i,j,k from the source; $C_{bi,j,k}$ is the conductance between the external source and cell i,j,k; $h_{bi,j,k}$ is the head assigned to the external source; and $h_{i,j,k}$ is the head in cell i,j,k. The relationship between cell i,j,k and the external source is shown schematically in figure 44. The constant-head source is represented by the apparatus on the right in figure 44, which holds the source head at the level h_b regardless of other factors; the link between the source and cell i,j,k is represented by the block of porous material $C_{bi,j,k}$. Note that figure 44 shows no mechanism to limit flow in either direction as $h_{i,j,k}$ rises or falls.

A graph of $Q_{bi,j,k}$ versus $h_{i,j,k}$ as given by equation (78) is shown in figure 45. In contrast to the River, Drain and ET Packages, the GHB Package provides no limiting value of flow to bound the linear function in either direction; and as the head difference between cell i,j,k and the source increases, flow into or out of the cell continues to increase without



Figure 44.—Schematic diagram illustrating principle of general-head boundary package.



Figure 45.—Plot of flow, Q_b, from a general-head boundary source into a cell as a function of head, h, in the cell where h_b is the source head.

limit. Care must accordingly be used in utilizing the GHB Package to insure that unrealistic flows into or out of the system do not develop during the course of simulation.

Because $Q_{bi,j,k}$ of equation (78) is defined as an inflow to the aquifer it must be added to the left side of equation (24). In terms of the expressions HCOF and RHS of equation (26), this is accomplished in the model by subtracting the term $C_{bi,j,k}$ from $HCOF_{i,j,k}$ and subtracting the term $C_{bi,j,k}h_{bi,j,k}$ from $RHS_{i,j,k}$ as the matrix equations are assembled.

General-Head Boundary Package Input

Input for the General-Head Boundary (GHB) Package is read from the unit specified in IUNIT(7).

FOR EACH SIMULATION

GHB1AL

1. Data: MXBND IGHBCB
   Format: I10 I10

FOR EACH STRESS PERIOD

GHB1RP

2. Data: ITMP
   Format: I10

3.	Data:	Layer	Row	Column	Boundary Head	Cond
	Format:	I10	I10	I10	F10.0	F10.0

(Input item 3 normally consists of one record for each GHB. If ITMP is negative or zero, item 3 is not read.)

Explanation of Fields Used in Input Instructions

MXBND--is the maximum number of general-head boundary cells at one time.

IGHBCB--is a flag and a unit number.

If IGHBCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If IGHBCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IGHBCB < 0, boundary leakage for each cell will be printed whenever ICBCFL is set.

ITMP--is a flag and a counter.

If ITMP < 0, GHB data from the preceding stress period will be reused.

If ITMP $\geq$ 0, ITMP is the number of general-head boundaries during the current stress period.

Layer--is the layer number of the cell affected by the head-dependent boundary.

Row--is the row number of the cell affected by the head-dependent boundary.

Column--is the column number of the cell affected by the head-dependent boundary.

Boundary head--is the head on the boundary.

Cond--is the hydraulic conductance of the interface between the aquifer cell and the boundary.

SAMPLE INPUT TO THE GENERAL HEAD BOUNDARY PACKAGE

DATA ITEM	EXPLANATION	INPUT RECORDS				
1	{MXBND, IGHBCB}_____	6	24			
2	STRESS PERIOD 1 {ITMP}_____	4				
3	FIRST BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	5	6	235.0	.0012
3	SECOND BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	4	6	230.0	.0012
3	THIRD BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	5	8	250.0	.0018
3	FOURTH BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	7	6	235.0	.0012
2	STRESS PERIOD 2 {ITMP}_____	-1				
2	STRESS PERIOD 3 {ITMP}_____	-1				
2	STRESS PERIOD 4 {ITMP}_____	6				
3	FIRST BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	5	6	235.0	.0012
3	SECOND BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	4	6	230.0	.0012
3	THIRD BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	5	8	250.0	.0018
3	FOURTH BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	7	6	235.0	.0018
3	FIFTH BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	9	6	235.0	.0012
3	SIXTH BOUNDARY {Layer, Row, Column, Head, Conductance_____	2	10	6	250.0	.0012